GRACe tutorial

June 30, 2017

In this tutorial we introduce the GRACe DSL and demonstrate how to define GRACe components and write programs using those components.

GRACe is divided into two modules, GCM and CP. The GCM module allows the user to define GRACeFUL Concept Map components and describe how they are connected. The CP module contains primitives for constructing constraint programs which describe the behavior of an individual component.

1 GCM components and ports

We model a `GCM` component by defining the ports it exposes.

A `Port` contains a value that can be constrained. Two ports can be linked together to describe the connection between their respective components. Information contained in a component that we want to access in other parts of our program can be exposed through the component's ports.

For instance, consider the following definition of a component which models a fixed amount of rain falling from the sky. It is parametrized on the amount of rain and has a port to expose that value to another component.

```
rain :: Float -> GCM (Port Float)
rain amount = do
  p <- createPort
  set p amount
  return p
```

Since the GRACe language is monadic we can use **do** notation and define our component in a sequential manner.

The `createPort` command creates a new port, and the `set` command constrains the value of the port `p` to be equal to `amount`.

2 CP

The `CP` module of GRACe supports reasoning about integer and floating-point arithmetic, boolean expressions, and arrays.

Computations in `CP` can be embedded in `GCM` using the `component` primitive. In this way we can embed constraints on a component's ports in the definition of the component.

Consider a `GCM` component representing a pump parametrised over the maximum flow through the pump:

```
pump :: Float -> GCM (Port Float, Port Float)
pump maxCap = do
  inPort  <- createPort
  outPort <- createPort
  component $ do                -- This is in CP
    inflow  <- value inPort
    outflow <- value outPort
    assert $  inflow === outflow
    assert $  inflow `inRange` (0, lit maxCap)
  return (inPort, outPort)
```

We define ports for the inflow and outflow, assert that their values must be equal, and that their values cannot exceed the pump's maximum capacity.

The `value` command reads the value from a port, and the `assert` command allows us to express constraints.

Note that we need to use `lit` to lift `maxCap`, which is a value in the host language Haskell, into the embedded language GRACe.

Finally we show a more complicated component, a water runoff area with an `inflow`, an `outlet` to which we may connect e.g. a pump, and an `overflow`.

Here we can see various different kinds of constraints that are supported by the `CP` module.

```
runoffArea :: Float -> GCM (Port Float, Port Float, Port Float)
runoffArea cap = do
  inflow   <- createPort
  outlet   <- createPort
  overflow <- createPort
  component $ do
    currentStored <- createVariable
    inf <- value inflow
    out <- value outlet
    ovf <- value overflow
    sto <- value currentStored
    assert $ sto === inf - out - ovf
    assert $ sto `inRange` (0, lit cap)
    assert $ (ovf .> 0) ==> (sto === lit cap)
    assert $ ovf .>= 0
  return (inflow, outlet, overflow)
```

3 GRACe programs

In a GRACe program we define instances of available components and define their connections by linking their ports.

As an example, here is a small GRACe program using the components defined earlier. We can think of it as modeling a rain runoff area, like a town square, which has been provided with a pump to alleviate possible flooding issues.

```
example :: GCM ()
example = do
```

```
  (inflowP, outflowP) <- pump 3
  (inflowS, outletS, overflowS) <- runoffArea 5
  rainflow <- rain 10

  link inflowP outletS
  link inflowS rainflow

  output overflowS "Overflow"
```

The `link` command links two ports together and asserts that their values must be equal.

The `output` command lets us inspect the resulting value at a `Port` after all constraints have been solved.

4 Example: Vegetable Oil Production

As an example we consider a simple optimization problem.

We have a set amount of farmland and three available crops, and would like to know much of each crop to grow on the land to maximize our vegetable oil production.

Each crop has parameters that state the yield of the crop, in tons, from one hectare of growing land, the amount of water required per hectare to grow the crop, and how much oil can be produced from one ton of the crop.

We define a component for each crop, using these parameters, with ports expressing the number of hectares, the oil yield, and the water consumption.

```
-- | GCM component for a single crop.
--
-- The component is parametrized on the crop's parameters and computes the
-- oil yield (in l) and water consumption (in Ml), given that we grow
-- so-and-so many ha of this crop.
crop :: CropParams -> GCM (Port Area, Port Water, Port Oil)
crop (y,w,o) = do
  -- Area (in ha) used to grow crop.
  areaPort  <- createPort

  -- Amount of water used by crop.
  waterPort <- createPort
  -- Amount of oil produced from crop.
  oilPort   <- createPort

  -- Constrain the values at the ports.
  component $ do
    areaValue  <- value areaPort
    oilValue   <- value oilPort
    waterValue <- value waterPort

    -- Calculate values from data.
    assert $ oilValue   === lit y * lit o * areaValue
    assert $ waterValue === lit w * areaValue
```

```haskell

  return (areaPort, waterPort, oilPort)
```

We also define components for the available farmland and water supply, and the oil production, which all have ports to link to each crop. We parametrize them on the number of different crops for the sake of generality.

```haskell
-- | GCM component for farmland.
--
-- The component is parametrized on the available amount of land (in ha)
-- and the number of different crops available to grow, and has ports
-- describing how the land is divided between the crops.
farm :: Area -> Int -> GCM [Port Area]
farm land numCrops = do
  -- Create a port for each crop.
  areaPorts <- mapM (\_ -> createPort) (take numCrops (repeat 0))
  component $ do
    areaVals <- sequence [value ap | ap <- areaPorts]
    --  The total area of crops is non-negative and is bounded by the available
    --  farmland. Each crop area is also non-negative.
    assert $ sum areaVals `inRange` (0, lit land)
    mapM_ (\x -> assert $ 0 .<= x) areaVals
  return areaPorts

-- | GCM component for water usage.
--
-- The component is parametrized on the available amount of water (in Ml)
-- and the number of different crops available to grow, and has ports
-- describing how the water is divided between the crops.
reservoir :: Water -> Int -> GCM [Port Water]
reservoir waterSource numCrops = do
  -- Create a port for each crop.
  waterPorts <- mapM (\_ -> createPort) (take numCrops (repeat 0))
  component $ do
    waterVals <- sequence [value wp | wp <- waterPorts]
    -- The total amount of water used is non-negative and is bounded by the
    -- available water reservoir. The amount for each crop is also non-negative.
    assert $ sum waterVals `inRange` (0, lit waterSource)
    mapM_ (\x -> assert $ 0 .<= x) waterVals
  return waterPorts

-- | GCM component for oil production.
--
-- The component is parametrized on the number of different crops available
-- to grow, and has a list of ports describing how much oil is produced by
-- each crop as well as a port containing the total amount of oil produced.
oilProduction :: Int -> GCM ([Port Oil], Port Oil)
oilProduction numCrops = do
  -- Create a port for each crop.
  oilCrops <- mapM (\_ -> createPort) (take numCrops (repeat 0))
```

```haskell
  oilOut <- createPort
  component $ do
    oilProduced <- value oilOut
    oilSources <- mapM value oilCrops
    -- The total amount of oil is the sum of the amounts from each crop.
    assert $ oilProduced === sum oilSources
  return (oilCrops, oilOut)
```

Our goal is to maximize the amount of oil produced, and we define the helper function `maximize` to help us express this:

```haskell
maximize :: Port Int -> GCM ()
maximize p = do
  g <- createGoal
  link p g
```

The command `createGoal` instantiates a GCM goal, which the constraint solver will attempt to maximize. Using this helper function we can simply write

```haskell
maximize p
```

to state that we would like to maximize the value at port `p`.

Conversely, if our goal is to minimize a certain value we can define a similar helper function `minimize`

```haskell
minimize :: Port Int -> GCM ()
        minimize p = do
          g <- createGoal
          linkBy (fun negate) p g
```

While the `link` command asserts that two values must be equal, the `linkby` command takes a function as a parameter to express a more complex constraint on the values.

The full code for the vegetable oil example can be seen in Appendix A

A Vegetable Oil Production - Full code

```haskell
module Main where

import CompileO (runGCM)
import GCM      ( GCM, output, component, createGoal
                , Port, createPort, link, value
                )
import CP       ( assert, lit, (===), (.<=), inRange )


-- * Vegetable oil manufacturing
-------------------------------------------------------------------------------
-- We define a system to describe growing and producing vegetable oil from
-- different types of crops.

-- | We use type synonyms to keep track of the different resources
```

```haskell
--   we are working with.
--
-- Farmland area is measured in ha
type Area  = Int
-- Crop yield is measure in t/ha
type Yield = Int
-- Water is measured in Ml
type Water = Int
-- Oil is measured in l
type Oil   = Int

-- Each crop has parameters describing its yield in t/ha,
-- its water demand in Ml/ha, and its oil content in l/t.
type CropParams = (Yield, Water, Oil)

-- | GCM component for a single crop.
--
-- The component is parametrized on the crop's parameters and computes the
-- oil yield (in l) and water consumption (in Ml), given that we grow
-- so-and-so many ha of this crop.
crop :: CropParams -> GCM (Port Area, Port Water, Port Oil)
crop (y,w,o) = do
  -- Area (in ha) used to grow crop.
  areaPort  <- createPort

  -- Amount of water used by crop.
  waterPort <- createPort
  -- Amount of oil produced from crop.
  oilPort   <- createPort

  -- Constrain the values at the ports.
  component $ do
    areaValue  <- value areaPort
    oilValue   <- value oilPort
    waterValue <- value waterPort

    -- Calculate values from data.
    assert $ oilValue   === lit y * lit o * areaValue
    assert $ waterValue === lit w * areaValue

  return (areaPort, waterPort, oilPort)

-- | GCM component for farmland.
--
-- The component is parametrized on the available amount of land (in ha)
-- and the number of different crops available to grow, and has ports
-- describing how the land is divided between the crops.
farm :: Area -> Int -> GCM [Port Area]
farm land numCrops = do
  -- Create a port for each crop.
```

```
  areaPorts <- mapM (\_ -> createPort) (take numCrops (repeat 0))
  component $ do
    areaVals <- sequence [value ap | ap <- areaPorts]
    -- The total area of crops is non-negative and is bounded by the available
    -- farmland. Each crop area is also non-negative.
    assert $ sum areaVals `inRange` (0, lit land)
    mapM_ (\x -> assert $ 0 .<= x) areaVals
  return areaPorts

-- | GCM component for water usage.
--
-- The component is parametrized on the available amount of water (in Ml)
-- and the number of different crops available to grow, and has ports
-- describing how the water is divided between the crops.
reservoir :: Water -> Int -> GCM [Port Water]
reservoir waterSource numCrops = do
  -- Create a port for each crop.
  waterPorts <- mapM (\_ -> createPort) (take numCrops (repeat 0))
  component $ do
    waterVals <- sequence [value wp | wp <- waterPorts]
    -- The total amount of water used is non-negative and is bounded by the
    -- available water reservoir. The amount for each crop is also non-negative.
    assert $ sum waterVals `inRange` (0, lit waterSource)
    mapM_ (\x -> assert $ 0 .<= x) waterVals
  return waterPorts

-- | GCM component for oil production.
--
-- The component is parametrized on the number of different crops available
-- to grow, and has a list of ports describing how much oil is produced by
-- each crop as well as a port containing the total amount of oil produced.
oilProduction :: Int -> GCM ([Port Oil], Port Oil)
oilProduction numCrops = do
  -- Create a port for each crop.
  oilCrops <- mapM (\_ -> createPort) (take numCrops (repeat 0))
  oilOut <- createPort
  component $ do
    oilProduced <- value oilOut
    oilSources <- mapM value oilCrops
    -- The total amount of oil is the sum of the amounts from each crop.
    assert $ oilProduced === sum oilSources
  return (oilCrops, oilOut)

-- | In our example problem we have 3 crops: Soybeans, sunflower seeds and
--   cotton seeds, parametrized by the following table:
--
--   Crop            Yield [t/ha]   Water demand [Ml/ha]  Oil content [l/t]
--   ~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~
--   Soybeans         3              5                     178
--   Sunflower seeds  2              4                     216
```

```haskell
--   Cotton seeds      1               1                      433
--
-- Note the units.

-- | We define a type for our crops and a function to keep track of
--   parameters
data Crop = Soy | Sunflower | Cotton
cropTable :: Crop -> (Yield, Water, Oil)
cropTable Soy       = (3, 5, 178)
cropTable Sunflower = (2, 4, 216)
cropTable Cotton    = (1, 1, 433)

-- | Help function for maximizing goal.
maximize :: Port Int -> GCM ()
maximize p = do
  g <- createGoal
  link p g

-- | GCM program for optimizing the area of land on which to grow each of the 3
--   available crops in order to maximize oil production.
--
--   In this example the following quantities are available:
--   Farmland: 1,600 ha
--   Water:    5,000 Ml
--
problem :: GCM ()
problem = do
  -- Create system
  (soyArea, soyWater, soyOil) <- crop $ cropTable Soy
  (sunArea, sunWater, sunOil) <- crop $ cropTable Sunflower
  (cotArea, cotWater, cotOil) <- crop $ cropTable Cotton

  [soy_a, sun_a, cot_a] <- farm 1600 3
  [soy_w, sun_w, cot_w] <- reservoir 5000 3
  ([soy_o, sun_o, cot_o], oilProduced) <- oilProduction 3

  -- Link the appropriate ports together.
  link soyArea soy_a
  link soyWater soy_w
  link soyOil soy_o

  link sunArea sun_a
  link sunWater sun_w
  link sunOil sun_o

  link cotArea cot_a
  link cotWater cot_w
  link cotOil cot_o

  -- Our goal is to maximize the amount of oil produced.
```

```
  maximize oilProduced

  -- We print the values we would like to see.
  output oilProduced "Oil produced"
  output soyArea "Soybean area"
  output sunArea "Sunflower area"
  output cotArea "Cotton area"

main :: IO ()
main = print =<< runGCM problem
```